Behavior Research Methods & Instrumentation
1980, Vol. 12 (1), 58-62

A computer-controlled cardiotachometer

WOLFGANG KLOSTERHALFEN
University of Düsseldorf, 4000 Düsseldorf, Federal Republic of Germany

A versatile display system for presenting heart rate information is described. The display has a table with 10 rows and 50 columns. This 10 by 50 matrix contains all integers from 0 to 499, so that it can display the instantaneous R-R intervals of a rat's EKG to the nearest millisecond. At each beat 1 of 10 vertically and 1 of 50 horizontally arranged indicator lights are turned on to provide the coordinates for reading the table. Either beat-to-beat or averaged heart rate or period is displayed. Small and large changes can be easily and accurately observed on the same scale. The lights are controlled by a PDP-8/e. A listing of the assembler program is presented.

In a large number of psychophysiological investigations involving measurement of heart rate (HR), an immediate indication of ongoing HR behavior is essential or at least helpful (e.g., studies of biofeedback, stress, habituation, or classical conditioning).

TYPES OF CARDIOTACHOMETERS

Several different cardiotachometers are available. The design of these units is usually based on analog circuitry that provides a voltage proportional to the reciprocal of the time interval between successive R waves. HR is indicated by a needle or pen that points to a scale (voltmeter type) or writes "staircases" on recording paper (tachograph) (e.g., Pope, Deboo, & Smith, 1968). Readings from these instruments are far from being exact.

Digital tachometers are highly accurate (e.g., Elings & Holly, 1973), but the dynamics of HR changes are poorly represented by numbers, and watching beat-to-beat changes in HR over a longer period of time on a LED display is a tedious job that absorbs most of the experimenter's attention.

Computers have been used successfully for the interpretation of the EKG complex (e.g., Wartak, Milliken, & Karchmar, 1970), but they have contributed little to improving HR monitoring. A computer-controlled cardiotachometer (CCC) that is accurate, easily readable, and versatile is described here.

COMPUTER-CONTROLLED CARDIOTACHOMETER

Display

The CCC's display has an interchangeable table of HR or heart period values. Ten vertically and 50 horizontally arranged indicator lights provide the coordinates for reading this 10 by 50 table (see Figure 1): At each beat one of the vertical and one of the horizontal lights are turned on. The first row of this matrix may, for example, contain R-R intervals from 0 (upper left corner) to 49 msec (upper right corner). For these values the top light of the vertical string and one of the horizontal lights point to the corresponding number in the table. The second verticle light is used for the second row (50 to 99), and so on, so that the lights can indicate any integer from 0 to 499.

The table shown in Figure 1, which is used for monitoring HR in rats, is based on this scaling. But following a convention, values are expressed in beats per minute (bpm).

Computer

A PDP-8/e computer with an almost noiseless reed relay interface (special fabrication) and a real-time clock (DK8-ES) was employed to control the CCC. However, almost any microcomputer with real-time clock and (multiplexed) TTL outputs to drive LEDs should be satisfactory (e.g., Brown & Deffenbacher, 1978). The EKG signal is fed directly or via an R-wave detector circuit (Shimizu, 1978) into the clock's Schmitt trigger input. An assembler program written in PAL8 uses approximately 100 locations (see Appendix). It measures R-R intervals and sets the relays for the lights of the CCC (see Figure 2). The 50 horizontal lights are multiplexed: They are connected to power in groups of 5 by 10 relays of Register 1, and grounded in groups of 10 by 5 relays of Register 2; 50 diodes (omitted in Figure 2) are connected in series to prevent loops. The 10 vertical lights are turned on by 10 relays of Register 3.

Depending on the position of bit 0 of the switch register, the CCC displays either beat-to-beat or averaged HR.

Applications

The CCC has operated reliably for several years and

Development of this system was supported in part by Grant Li 74/10 from the Deutsche Forschungsgemeinschaft to Gustav A. Lienert. Requests for reprints should be addressed to Wolfgang Klosterhalfen, Institut fuer Medizinische Psychologie der Universitaet Duesseldorf, Universitaetsstrasse 1, 4000 Duesseldorf, Federal Republic of Germany.

0005-7878/80/010058-05$00.75/0



Figure 1. Prototype of a computer-controlled cardiotachometer (100 x 30 x 8 cm) indicating 375 beats per minute.



Figure 2. Relay circuits for control of the tachometer's lights.

has proved very helpful in studying the effects of acoustic stimulation on HR in rats. Using the table shown in Figure 1 and an average over four intervals, the lights performed fairly continuous movements rather than jumps. The green, yellow, and red horizontal lights, indicating ranges of 241-300, 302-400, and 402-600 bpm, respectively, required little attention. As each range is spatially represented by a meter, rather than by a centimeter as in conventional tachographs, good HR readings can be obtained at a glance, even from several meters away.

The program for the CCC is short and does not require fast serving. It may be easily integrated into larger EKG programs. The CCC can be adapted easily to meet the experimenter's special needs. Changing its range, the time spacing between lights, or the number of averaged intervals can be done by changing the table and a few instructions in the program.

Principally, any variable changing over time may be displayed. Because of its large range, the instrument is well suited, for example, to monitor skin conductance or resistance levels. In this case, other than the table, only the program's input routine has to be changed to accept data from an A/D converter (e.g., AD8-EA).

REFERENCES

BROWN, E. L., & DEFFENBACHER, K. Microcomputers big and little: Selecting a low-cost laboratory computer network. *Behavior Research Methods & Instrumentation*, 1978, **10**, 241-245.

ELINGS, V., & HOLLY, D. A cardiotachometer which calculates rate digitally. *IEEE Transactions on Biomedical Engineering*, November 1973, 468-470.

POPE, J. M., DEBOO, A. J., & SMITH, D. B. D. A cardiotachometer with linear indication of beat-to-beat frequency. *Psychophysiology*, 1968, **4**, 486-492.

SHIMIZU, H. Reliable and precise identification of R-waves in the EKG with a simple peak detector. *Psychophysiology*, 1978, **15**, 499-501.

WARTAK, J., MILLIKEN, J. A., & KARCHMAR, J. Computer program for pattern recognition of electrocardiograms. *Computers and Biomedical Research*, 1970, **4**, 344-374.

Appendix
Listing of a PAL8 Assembler Program to Control the Cardiotachometer

```
                    / DEFINITIONS

                    CLSK=6131               / SKIP ON CLOCK INTERRUPT
                    CLOE=6132               / SET CLOCK ENABLE REGISTER PER AC
                    CLBA=6136               / CLOCK BUFFER TO AC
                    LDR1=6141               / LOAD RELAY REGISTER 1 WITH AC
                    LDR2=6143               / LOAD RELAY REGISTER 2 WITH AC
                    LDR3=6145               / LOAD RELAY REGISTER 3 WITH AC

                    / ROUTINE TO SET THE CLOCK ENABLE REGISTER:
                    / ON EACH SCHMITT TRIGGER INPUT THE CLOCK
                    / - TRANSFERS THE CLOCK COUNTER CONTENTS TO THE CL. BUFFER
                    / - CLEARS THE CLOCK COUNTER
                    / - AND STARTS COUNTING FROM ZERO AT 1 KHZ

                    *200
0200  7200  CLOCK,   CLA                   / CLEAR ACCUMULATOR
0201  1322          TAD ENABLE            / GET CLOCK CONTROL WORD
0202  6132          CLOE                  / LOAD IT INTO CLOCK ENABLE REGISTER

                    / ROUTINE TO MEASURE THE INTERVAL BETWEEN THE LAST TWO
                    / SCHMITT TRIGGER INPUTS

0203  6131  INPUT,  CLSK                  / HAS A SCHMITT TRIGGER FIRED ?
0204  5203          JMP .-1               / NO:  CHECK AGAIN
0205  6136          CLBA                  / YES: READ AND
0206  3323          DCA RRINT             /      SAVE CLOCK BUFFER CONTENTS

                    / ROUTINE TO CHOICE BETW. AVERAGED AND BEAT-TO-BEAT DISPLAY

0207  7604  SWITCH, LAS                   / LOAD AC WITH SWITCH REGISTER
0210  7500          SMA                   / IS SWITCH 0 SET ?
0211  5234          JMP INIT              / NO: PROVIDE A BEAT-TO-BEAT DISPLAY

                    / ROUTINE TO CALCULATE A MOVING AVERAGE

0212  7200  MOVAVR, CLA                   / YES: CALCULATE MOVING AVERAGE
0213  1326          TAD RRINT3            / STORE THE
0214  3327          DCA RRINT4            / FOUR
0215  1325          TAD RRINT2            / MOST
0216  3326          DCA RRINT3            / RECENT
0217  1324          TAD RRINT1            / R-R INTERVALS
0220  3325          DCA RRINT2            / IN
0221  1323          TAD RRINT             / CHRONOLOGICAL
0222  3324          DCA RRINT1            / ORDER,
0223  1327          TAD RRINT4            / DO
```

```
0224  1326          TAD RRINT3      / ADDITION
0225  1325          TAD RRINT2      / OF
0226  1324          TAD RRINT1      / LAST FOUR INTERVALS,
0227  7110          CLL RAR         / DIVIDE SUM
0230  7110          CLL RAR         / BY 4
0231  7430          SZL             / AND STORE
0232  7001          IAC             / AVERAGED
0233  3323          DCA RRINT       / R-R-INTERVAL


              / ROUTINE TO RESET CONTROL WORDS AND COUNTERS

0234  7200  INIT,   CLA             / CLEAR ACCUMULATOR
0235  1341          TAD K4000       / INITIALIZE
0236  3330          DCA HORPWG      / CONTROL
0237  1341          TAD K4000       / WORDS
0240  3331          DCA HORGRP      / FOR
0241  1341          TAD K4000       / RELAY
0242  3332          DCA VRTPOS      / REGISTERS
0243  1337          TAD M10         / AND
0244  3333          DCA PWGM10      / RESET
0245  1340          TAD M5          / COUNTERS
0246  3334          DCA GRPM5       / FOR
0247  1337          TAD M10         / RELAY
0250  3335          DCA VRTM10      / SELECTION
0251  1323          TAD RRINT       / SET UP COUNTER
0252  7040          CMA             / ACCORDING TO
0253  3336          DCA COUNTR      / MILLISECONDS ELAPSED

              / ROUTINE TO DETERMINE NUMBER OF LOOPS

0254  2336  LOOP,   ISZ COUNTR      / COUNTER = 0 ?
0255  7410          SKP             / NO:  MODIFY CONTROL WORD(S)
0256  5310          JMP LOADRG      / YES: LOAD RELAY REGISTERS

              / ROUTINE TO DETERMINE A LIGHT'S POSITION WITHIN A HORIZON-
              / TAL GROUP: ONE ROTATION FOR EVERY MILLISECOND

0257  7300  MSEC01, CLA CLL         / CLEAR AC AND LINK
0260  1330          TAD HORPWG      / GET RELAY CONTROL WORD 1
0261  7010          RAR             / ROTATE AC AND LINK RIGHT
0262  3330          DCA HORPWG      / SAVE RELAY CONTROL WORD 1
0263  2333          ISZ PWGM10      / 10. ROTATION ?
0264  5254          JMP LOOP        / NO:  LOOP BACK
0265  1337          TAD M10         / YES: RESTORE PWGM10
0266  3333          DCA PWGM10      / AS COUNTER
0267  1341          TAD K4000       / CLEAR BIT 10 AND SET BIT 3
0270  3330          DCA HORPWG      / OF RELAY CONTROL WORD 1

              / ROUTINE TO SELECT ONE OF FIVE HORIZONTAL GROUPS;
              / ONE ROTATION FOR EVERY 10 MILLISECONDS

0271  1331  MSEC10, TAD HORGRP      / GET RELAY CONTROL WORD 2
0272  7010          RAR             / ROTATE AC AND LINK RIGHT
0273  3331          DCA HORGRP      / SAVE RELAY CONTROL WORD 2
0274  2334          ISZ GRPM5       / 5. ROTATION ?
0275  5254          JMP LOOP        / NO:  LOOP BACK
```

```
0276  1340          TAD M5          / YES: RESTORE GRPM5
0277  3334          DCA GRPM5       / AS COUNTER
0300  1341          TAD K4000       / CLEAR BIT 10 AND SET BIT 0
0301  3331          DCA HORGRP      / OF RELAY CONTROL WORD 2
                / ROUTINE TO DETERMINE A LIGHT'S POSITION WITHIN THE
                / VERTICAL STRING: ONE ROTATION FOR EVERY 50 MSECS

0302  1332  MSEC50, TAD VRTPOS      / GET RELAY CONTROL WORD 3
0303  7010          RAR             / ROTATE AC AND LINK RIGHT
0304  3332          DCA VRTPOS      / SAVE RELAY CONTROL WORD 3
0305  2335          ISZ VRTM10      / 10. ROTATION ?
0306  5254          JMP LOOP        / NO:  LOOP BACK
0307  5310          JMP LOADRG      / YES: OUT OF RANGE; NO VERT. LIGHT

                / ROUTINE TO CLOSE THREE RELAYS

0310  7200  LOADRG, CLA             / SELECT AND CLOSE
0311  1330          TAD HORPWG      / ONE RELAY
0312  6141          LDR1            / IN RELAY REGISTER 1
0313  7200          CLA             / SELECT AND CLOSE
0314  1331          TAD HORGRP      / ONE RELAY
0315  6143          LDR2            / IN RELAY REGISTER 2
0316  7200          CLA             / SELECT AND CLOSE
0317  1332          TAD VRTPOS      / ONE RELAY
0320  6145          LDR3            / IN RELAY REGISTER 3
0321  5203          JMP INPUT       / RETURN TO WAITING LOOP


0322  3307  ENABLE, 3307            / CLOCK CONTROL WORD
0323  0000  RRINT,  0000            / LAST R-R INTERVAL IN MILLISECONDS
0324  0000  RRINT1, 0000            / TEMPORARY
0325  0000  RRINT2, 0000            / STORAGE
0326  0000  RRINT3, 0000            / OF LAST
0327  0000  RRINT4, 0000            / R-R INTERVALS
0330  0000  HORPWG, 0000            / RELAY CONTROL WORD 1
0331  0000  HORGRP, 0000            / RELAY CONTROL WORD 2
0332  0000  VRTPOS, 0000            / RELAY CONTROL WORD 3
0333  0000  PWGM10, 0000            / ROTATION COUNTER 1
0334  0000  GRPM5,  0000            / ROTATION COUNTER 2
0335  0000  VRTM10, 0000            / ROTATION COUNTER 3
0336  0000  COUNTR, 0000            / LOOP COUNTER
0337  7766  M10,    7766            / DECIMAL -10
0340  7773  M5,     7773            / DECIMAL -5
0341  4000  K4000,  4000            / OCTAL 4000 (BIT 0 = 1)
      $
```

```
CLDA   6136    CLOCK  0200    CLOE   6132    CLSK   6131    COUNTR 0336
ENABLE 0322    GRPM5  0334    HORGRP 0331    HORPWG 0330    INIT   0234
INPUT  0203    K4000  0341    LDR1   6141    LDR2   6143    LDR3   6145
LOADRG 0310    LOOP   0254    MOVAVR 0212    MSEC01 0257    MSEC10 0271
MSEC50 0302    M10    0337    M5     0340    PWGM10 0333    RRINT  0323
RRINT1 0324    RRINT2 0325    RRINT3 0326    RRINT4 0327    SWITCH 0207
VRTM10 0335    VRTPOS 0332
```

(Received for publication June 27, 1979;
revision accepted November 15, 1979.)